The code below will plot a Slope Field for the differential equation

$$\frac{dy}{dx} = f(x, y).$$

For the purposes of demonstration, we pick:

$$f(x, y) = x^2 y - xy^2.$$

```
f[x_, y_] := x^2 * y - x * y^2
```

The command for producing the Slope Field is VectorPlot, and some sample code is given below.

```
VectorPlot[{1, f[x, y]}, {x, -5, 5}, {y, -5, 5}, Axes → True, FrameLabel → {x, y},
 VectorPoints → 25, VectorStyle → Black, VectorScale → {0.025, 0.001, None}]
```



The only required arguments for VectorPlot are the first three which respectively specify the differential equation, the range of x-values for the plot, and the range of y-values for the plot. The remaining options give *Mathematica* details about how to render the plot of the Slope FIeld (you should try varying these commands to see how what effect they have on the output). The first two optional commands tell

Mathematica to include the axes and label them x and y. The VectorPoints option tells *Mathematica* how many points to plot slopes for (in this case it picks 25 x-values and 25 y-values in the specified ranges to give $25^2 = 625$ points total). The VectorStyle command is useful for various purposes, but here we have only used it to specify the color of the slope lines. The final option is more complicated. VectorScale tells *Mathematica* how long each slope line should be and whether to draw arrowheads at the tip (which we do not want). You can play with these options to see their effect on the plot, but you can probably use these settings for any Slope Field you would like to draw.

To add sample solution curves to the plot, we add the following optional arguments to the VectorPlot command.

```
VectorPlot[{1, f[x, y]}, {x, -5, 5}, {y, -5, 5}, FrameLabel → {x, y}, Axes → True,
 VectorScale → {0.025, 0.001, None}, VectorPoints → 25, VectorStyle → Black,
 StreamPoints → Fine, StreamScale → None, StreamStyle → {{Thickness[0.003], Red}}]
```



Mathematica refers to these solution curves as stremalines. For the StreamPoints option, you can give Coarse, Fine, or an actual number. StreamScale controls arrowheads on the plot of the solution curves (which we do not want). Finally, StreamStyle controls several options for drawing the solution curves. Here, we choose the thickness of the curves and their color.

To find the general solution to the DE with *Mathematica*, we use the DSolve command.

```
DSolve[y'[x] == f[x, y[x]], y[x], x]
```

$$\left\{\left\{y[x] \to \frac{3\, e^{\frac{x^3}{3}} \left(-x^3\right)^{2/3}}{3 \left(-x^3\right)^{2/3} C[1] - 3^{2/3} x^2\, \text{Gamma}\left[\frac{2}{3}, -\frac{x^3}{3}\right]}\right\}\right\}$$

The first argument in the DSolve command is the DE we want to solve. The second argument specifies the dependent variable (here as a function y[x]), and the final argument gives the independent variable. Note that this DE is not one we know how to solve by hand (though *Mathematica* has no problem finding the answer)! If we want to solve an IVP, we bundle the DE and the initial value together as a list in the first argument. For example, if we would like the solution satisfying y(0) = 2, we use the following command.

```
DSolve[{y'[x] == f[x, y[x]], y[0] == 2}, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{6\, e^{\frac{x^3}{3}} \left(-x^3\right)^{2/3}}{-3 \left(-x^3\right)^{2/3} + 2 \left(-1\right)^{1/3} 3^{2/3} \left(-x^3\right)^{2/3} \text{Gamma}\left[\frac{2}{3}\right] + 2 \times 3^{2/3} x^2\, \text{Gamma}\left[\frac{2}{3}, -\frac{x^3}{3}\right]}\right\}\right\}$$

We can plot this function directly by pasting the output into a plot command:

$$\text{Plot}\left[-\frac{6\, e^{\frac{x^3}{3}} \left(-x^3\right)^{2/3}}{-3 \left(-x^3\right)^{2/3} + 2\, (-1)^{1/3}\, 3^{2/3} \left(-x^3\right)^{2/3}\, \text{Gamma}\left[\frac{2}{3}\right] + 2 \times 3^{2/3}\, x^2\, \text{Gamma}\left[\frac{2}{3}, -\frac{x^3}{3}\right]},\right.$$
$$\left.\{x, 0, 3\},\ \text{AxesLabel} \to \{x, y\}\right]$$